
	UNIDAD Nº :3

Recursos de ASP.Net.
	TEMAS: 

Imágenes dinámicas.
	Clase

10


Objetivos:

· Creación de imágenes dinámicas.

· Generación de distintos formatos gráficos..

Introducción 

Cuando debemos generar imágenes a partir de datos que se extraen de una base de datos o datos ingresados desde el navegador debemos construir la imagen en dicho momento.

La ASP.Net provee de clases orientadas a generar imágenes en memoria y su posterior materialización en un archivo.

Imágenes dinámicas
1 - Programa “Hola Mundo”

Para crear una imagen en forma dinámica debemos seguir una serie de pasos:

a - Debemos modificar el archivo webform agregando lo indicado en negrita:

<%@ Page ContentType="image/gif" Language="vb" AutoEventWireup="false" Codebehind="WebForm1.aspx.vb" Inherits="WebApplication27.WebForm1"%>

Al resto de las marcas Html debemos borrarlas ya que para generar una imagen no serán de utilidad. Es decir que en realidad no se trata de un webform lo que envía el servidor sino se trata de una imagen.

b – El archivo *.aspx.vb es:

'Importamos el espacio de nombre que declara la clase Bitmap, Graphics, Font y SolidBrush

Imports System.Drawing

'Importamos el espacio de nombre que declara la clase ImageFormat

Imports System.Drawing.Imaging

Partial Class pagina1

    Inherits System.Web.UI.Page

    Protected Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles Me.Load

        'Creamos un objeto de la clase Bitmap

        Dim mapabit As New Bitmap(500, 300)

        Dim lienzo As Graphics

        'Iicializamos la variable de tipo Graphics con el bitmap creado

        lienzo = Graphics.FromImage(mapabit)

        'Creamos una fuente

        Dim fuente1 As New Font("Arial", 40)

        'Creamos un pincel

        Dim pincel1 As New SolidBrush(Color.Red)

        'Graficamos dentro de la imagen

        lienzo.DrawString("Hola Mundo", fuente1, pincel1, 10, 10)

        'Enviamos la imagen al navegador que la solicitó

        mapabit.Save(Response.OutputStream, ImageFormat.Gif)

    End Sub

End Class

El resultado de la ejecución es:

Es importante tener en cuenta que cuando ejecutamos este webform en realidad lo que retorna es una imagen con formato gif y no un archivo HTML. Fácilmente podemos comprobarlo tratando de ver el código fuente de la página presionando el botón derecho del mouse dentro del navegador.

2 – Inserción de una imagen dinámica en un webform.

Para la incorporación de la imagen en una página propiamente dicha debemos hacer lo siguiente:

Creamos una página con un código similar a este:

<%@ Page Language="VB" AutoEventWireup="false" CodeFile="pagina1.aspx.vb" Inherits="pagina1" %>

<!DOCTYPE html PUBLIC "-//W3C//DTD XHTML 1.0 Transitional//EN" "http://www.w3.org/TR/xhtml1/DTD/xhtml1-transitional.dtd">

<html xmlns="http://www.w3.org/1999/xhtml">

<head runat="server">

    <title>Página sin título</title>

</head>

<body>

    <form id="form1" runat="server">

    <div> 

        <img alt="" src="imagen1.aspx" /></div>

    </form>

</body>

</html>

Es decir que con la marca HTML Img indicamos en la propiedad src el nombre del archivo que genera la imagen.

3 –Generación de un Captcha.

Captcha es el acrónimo de Completely Automated Public Turing text to tell Computers and Humans Apart (Prueba de Turing pública y automática para diferenciar a máquinas y humanos).

Se trata de una prueba desafío-respuesta utilizada en computación para determinar cuándo el usuario es o no humano.

Confeccionaremos una imagen con un número aleatorio entre 1000 y 9999 y para dificultar su lectura por un programa automático de lectura lo rayaremos con líneas de distintos colores.

Por último incorporaremos almacenaremos el valor en una variable de sesión y validaremos el valor que ingresa el usuario en un control de tipo textbox.

a -  Primero generamos un gráfico dinámico (captcha.aspx.vb y captcha.aspx):

Tener en cuenta que el archivo captcha.aspx solo debemos disponer (borramos todo el código HTML y en el bloque de definición de atributos aspx agregamos la propiedad ContentType indicando que generará una imagen en formato gif):

<%@ Page ContentType="image/gif" Language="VB" AutoEventWireup="false" CodeFile="captcha.aspx.vb" Inherits="captcha" %>

Luego el código de la clase será:

Imports System.Drawing

Imports System.Drawing.Imaging

Partial Class captcha

    Inherits System.Web.UI.Page

    Protected Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles Me.Load

        Dim mapabit As New Bitmap(100, 40)

        Dim lienzo As Graphics

        lienzo = Graphics.FromImage(mapabit)

        Dim nro As Integer

        nro = Rnd() * 10000

        Dim fuente1 As New Font("Arial", 22)

        Dim pincel1 As New SolidBrush(Color.Red)

        lienzo.DrawString(nro, fuente1, pincel1, 4, 4)

        'Dibujar lineas

        For f As Integer = 1 To 10

            Dim x1 As Integer = Rnd() * 100

            Dim y1 As Integer = Rnd() * 40

            Dim x2 As Integer = Rnd() * 100

            Dim y2 As Integer = Rnd() * 40

            Dim lapiz1 As New Pen(Color.FromArgb(Rnd() * 255, Rnd() * 255, Rnd() * 255))

            lienzo.DrawLine(lapiz1, x1, y1, x2, y2)

        Next

        Me.Session("codigo") = nro

        mapabit.Save(Response.OutputStream, ImageFormat.Gif)

    End Sub

End Class

Hacemos el import de los dos espacios de nombres:

Imports System.Drawing

Imports System.Drawing.Imaging

En el método Page_Load creamos un objeto de la clase BitMap indicando el ancho y el alto en píxeles.

Obtenemos una referencia de la imagen almacenándola en una variable de tipo Graphics.

        Dim lienzo As Graphics

        lienzo = Graphics.FromImage(mapabit)

Generamos un valor aleatorio:

        nro = Rnd() * 10000

Imprimimos el número dentro de la imagen:

        Dim fuente1 As New Font("Arial", 22)

        Dim pincel1 As New SolidBrush(Color.Red)

        lienzo.DrawString(nro, fuente1, pincel1, 4, 4)

Dibujamos 10 líneas de distintos colores en forma aleatoria tapando parcialmente el número aleatorio:

        For f As Integer = 1 To 10

            Dim x1 As Integer = Rnd() * 100

            Dim y1 As Integer = Rnd() * 40

            Dim x2 As Integer = Rnd() * 100

            Dim y2 As Integer = Rnd() * 40

            Dim lapiz1 As New Pen(Color.FromArgb(Rnd() * 255, Rnd() * 255, Rnd() * 255))

            lienzo.DrawLine(lapiz1, x1, y1, x2, y2)

        Next

Almacenamos el valor en una variable de sesión para poder recuperarlo en otra página y poder compararlo con el que ingresa el usuario por teclado:

      Me.Session("codigo") = nro

Enviamos la imagen al navegador que lo solicitó:

        mapabit.Save(Response.OutputStream, ImageFormat.Gif)

 b – Creamos un formulario web donde insertamos el Captcha (imagen) y disponemos además un control de tipo textbox y un button:

Debemos generar un webform semejante a este:

Para el evento Click del objeto Button1 disponemos el siguiente código:

    Protected Sub Button1_Click(ByVal sender As Object, ByVal e As System.EventArgs) Handles Button1.Click

        If Me.Session("codigo") = Me.TextBox1.Text Then

            Me.Label1.Text = "Ingresó el codigo correcto"

        Else

            Me.Label1.Text = "Incorrecto"

        End If

    End Sub
Luego mediante un if controlamos el valor del captcha y el valor ingresado por el usuario en el textbox1. Mostramos en un label el resultado de la  comparación.

4 –Gráfico de tarta.

Para implementar un gráfico de tarta debemos emplear el método fillpie de la clase Graphics. Desarrollaremos una aplicación que solicite en un webform tres valores enteros por teclado y llamaremos a un gráfico dinámico que mostrará un gráfico de tarta rescatando los valores del webform.

El webform envía mediante parámetros los datos ingresados.

El webform tiene la siguiente interfaz:

Y el código que debemos implementar para el evento Click del objeto Button es:

    Protected Sub Button1_Click(ByVal sender As Object, ByVal e As System.EventArgs) Handles Button1.Click

        Me.Response.Redirect("graficotarta.aspx?v1=" & Me.TextBox1.Text & "&v2=" & Me.TextBox2.Text & "&v3=" & Me.TextBox3.Text)

    End Sub
Es decir redireccionamos al archivo graficotarta.aspx pasando 3 parámetros en la cabecera de la llamada.

El archivo que genera la imagen dinámica (graficotarta.aspx.vb) es la siguiente:

Imports System.Drawing

Imports System.Drawing.Imaging

Partial Class graficotarta

    Inherits System.Web.UI.Page

    Protected Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles Me.Load

        Dim mapabit As New Bitmap(800, 600)

        Dim lienzo As Graphics

        lienzo = Graphics.FromImage(mapabit)

        Dim valor1 As Integer = Me.Request.QueryString("v1")

        Dim valor2 As Integer = Me.Request.QueryString("v2")

        Dim valor3 As Integer = Me.Request.QueryString("v3")

        Dim total As Integer = valor1 + valor2 + valor3

        Dim grados1 As Integer = valor1 / total * 360

        Dim pincel1 As New SolidBrush(Color.Red)

        lienzo.FillPie(pincel1, 100, 100, 400, 400, 0, grados1)

        Dim grados2 As Integer = valor2 / total * 360

        Dim pincel2 As New SolidBrush(Color.Blue)

        lienzo.FillPie(pincel2, 100, 100, 400, 400, grados1, grados2)

        Dim grados3 As Integer = valor3 / total * 360

        Dim pincel3 As New SolidBrush(Color.Green)

        lienzo.FillPie(pincel3, 100, 100, 400, 400, grados1 + grados2, grados3)

        'Referencias

        lienzo.FillRectangle(pincel1, 600, 500, 20, 20)

        Dim fuente As New Font("Arial", 10)

        lienzo.DrawString(valor1, fuente, pincel1, 630, 500)

        lienzo.FillRectangle(pincel2, 600, 530, 20, 20)

        lienzo.DrawString(valor2, fuente, pincel2, 630, 530)

        lienzo.FillRectangle(pincel3, 600, 560, 20, 20)

        lienzo.DrawString(valor3, fuente, pincel3, 630, 560)

        mapabit.Save(Response.OutputStream, ImageFormat.Gif)

    End Sub

End Class
También recordemos que el archivo (graficotarta.aspx) debemos borrar todo el contenido HTML y modificar la cabecera:

<%@ Page Language="VB" AutoEventWireup="false" CodeFile="graficotarta.aspx.vb" Inherits="graficotarta" contenttype="image/gif"%>
Veamos el algoritmo para mostrar el gráfico de tarta dentro de la imagen. En el método Load generamos la imagen. Creamos un objeto de la clase Bitmap y obtenemos la referencia al objeto de la clase Graphics contenido en el mismo:

        Dim mapabit As New Bitmap(800, 600)

        Dim lienzo As Graphics

        lienzo = Graphics.FromImage(mapabit)

Recuperamos los tres valores ingresados en el webform:

        Dim valor1 As Integer = Me.Request.QueryString("v1")

        Dim valor2 As Integer = Me.Request.QueryString("v2")

        Dim valor3 As Integer = Me.Request.QueryString("v3")

Calculamos la cantidad de grados que le corresponde a cada valor y dibujamos el trozo de tarta respectivo:

        Dim total As Integer = valor1 + valor2 + valor3

        Dim grados1 As Integer = valor1 / total * 360

        Dim pincel1 As New SolidBrush(Color.Red)

        lienzo.FillPie(pincel1, 100, 100, 400, 400, 0, grados1)

        Dim grados2 As Integer = valor2 / total * 360

        Dim pincel2 As New SolidBrush(Color.Blue)

        lienzo.FillPie(pincel2, 100, 100, 400, 400, grados1, grados2)

        Dim grados3 As Integer = valor3 / total * 360

        Dim pincel3 As New SolidBrush(Color.Green)

        lienzo.FillPie(pincel3, 100, 100, 400, 400, grados1 + grados2, grados3)

Dibujamos los recuadros de referencias con el valor que le corresponde a cada trozo de tarta:            

        lienzo.FillRectangle(pincel1, 600, 500, 20, 20)

        Dim fuente As New Font("Arial", 10)

        lienzo.DrawString(valor1, fuente, pincel1, 630, 500)

        lienzo.FillRectangle(pincel2, 600, 530, 20, 20)

        lienzo.DrawString(valor2, fuente, pincel2, 630, 530)

        lienzo.FillRectangle(pincel3, 600, 560, 20, 20)

        lienzo.DrawString(valor3, fuente, pincel3, 630, 560)

        mapabit.Save(Response.OutputStream, ImageFormat.Gif)

El resultado del gráfico es:

4 –Gráfico de barra horizontal.

Para implementar un gráfico de barra horizontal debemos emplear el método fillrect de la clase Graphics. Desarrollaremos una aplicación que solicite en un webform tres valores enteros por teclado y llamaremos a un gráfico dinámico que mostrará un gráfico de barra horizontal rescatando los valores del webform.

El webform envía mediante parámetros los datos ingresados.

El interfaz para la carga de los tres valores enteros en un webform es similar al problema del gráfico de tarta.

El código que debemos implementar para el evento Click del objeto Button es:

    Protected Sub Button1_Click(ByVal sender As Object, ByVal e As System.EventArgs) Handles Button1.Click

        Me.Response.Redirect("graficobarrahorizontal.aspx?v1=" & Me.TextBox1.Text & "&v2=" & Me.TextBox2.Text & "&v3=" & Me.TextBox3.Text)

    End Sub
Es decir redireccionamos al archivo graficobarrahorizontal.aspx pasando 3 parámetros en la cabecera de la llamada.

El archivo que genera la imagen dinámica (graficobarrahorizontal.aspx.vb) es la siguiente:

Imports System.Drawing

Imports System.Drawing.Imaging

Partial Class Default2

    Inherits System.Web.UI.Page

    Private Function RetornarMayor(ByVal x1 As Integer, ByVal x2 As Integer, ByVal x3 As Integer) As Integer

        If x1 > x2 And x1 > x3 Then

            Return x1

        Else

            If x2 > x3 Then

                Return x2

            Else

                Return x3

            End If

        End If

    End Function

    Protected Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs) Handles Me.Load

        Dim mapabit As New Bitmap(800, 400)

        Dim lienzo As Graphics

        lienzo = Graphics.FromImage(mapabit)

        Dim valor1 As Integer = Me.Request.QueryString("v1")

        Dim valor2 As Integer = Me.Request.QueryString("v2")

        Dim valor3 As Integer = Me.Request.QueryString("v3")

        Dim mayor As Integer = Me.RetornarMayor(valor1, valor2, valor3)

        Dim largo1 As Integer = valor1 / mayor * 400

        Dim pincel1 As New SolidBrush(Color.Red)

        lienzo.FillRectangle(pincel1, 50, 50, 50 + largo1, 80)

        Dim largo2 As Integer = valor2 / mayor * 400

        Dim pincel2 As New SolidBrush(Color.Green)

        lienzo.FillRectangle(pincel2, 50, 150, 50 + largo2, 80)

        Dim largo3 As Integer = valor3 / mayor * 400

        Dim pincel3 As New SolidBrush(Color.Blue)

        lienzo.FillRectangle(pincel3, 50, 250, 50 + largo3, 80)

        'Cantidades

        Dim pincel4 As New SolidBrush(Color.Black)

        Dim fuente1 As New Font("Arial", 30)

        lienzo.DrawString(valor1, fuente1, pincel4, 60, 80)

        lienzo.DrawString(valor2, fuente1, pincel4, 60, 160)

        lienzo.DrawString(valor3, fuente1, pincel4, 60, 260)

        mapabit.Save(Response.OutputStream, ImageFormat.Gif)

    End Sub

End Class

El método RotornarMayor recibe como parámetros tres enteros y retorna el mayor de los mismos:

    Private Function RetornarMayor(ByVal x1 As Integer, ByVal x2 As Integer, ByVal x3 As Integer) As Integer

        If x1 > x2 And x1 > x3 Then

            Return x1

        Else

            If x2 > x3 Then

                Return x2

            Else

                Return x3

            End If

        End If

    End Function

El método Load es el que crea el gráfico de barra horizontal propiamente dicho.

Para graficar cada barra:

        Dim largo1 As Integer = valor1 / mayor * 400

        Dim pincel1 As New SolidBrush(Color.Red)

        lienzo.FillRectangle(pincel1, 50, 50, 50 + largo1, 80)

Debemos dividir el valor correspondiente con respecto al mayor de los tres. Este valor como máximo puede ser 1. Luego multiplicamos este valor por 400 (este valor representa el largo de barra mayor)

Graficamos mediante el método FillRectangle.

Mostramos los valores enteros que les corresponde a cada barra:

        'Cantidades

        Dim pincel4 As New SolidBrush(Color.Black)

        Dim fuente1 As New Font("Arial", 30)

        lienzo.DrawString(valor1, fuente1, pincel4, 60, 80)

        lienzo.DrawString(valor2, fuente1, pincel4, 60, 160)

        lienzo.DrawString(valor3, fuente1, pincel4, 60, 260)

Ejercicios Propuestos
1 – Confeccionar un gráfico dinámico que genere un gráfico de barras verticales.

2 – Confeccionar un gráfico dinámico que genere una barra porcentual (mostrar los porcentajes que les corresponde a cada trozo fuera de la barra).

3 – Generar una imagen a partir de otra y agregarle un texto en forma dinámica. En algunos casos cuando se quiere proteger la propiedad de imágenes es aplicarles un texto que indica que la imagen no puede ser usada en otro sitio para fines comerciales. 

4 – Confeccionar un algoritmo que permita ingresar los datos personales de una persona y luego genere una tarjeta personal con dichos datos:

ej:. 

